Example 3b: ANOVA Models for the Practice Effects Data
(complete syntax, data, and output available for SAS, STATA, and R electronically)

This example comes from Hoffman (2015) chapter 3 (example 3b). We will be examining an outcome of response time (RT) in milliseconds across six practice sessions to a measure of processing speed (as measured by the number match 3 test) in a sample of 101 older adults. Of interest right now is the pattern of variance and covariance in RT across sessions. For an example results section, please see the end of chapter 3.

Later in Example 6 we will examine the extent to which individual differences in change in RT can be described by polynomial models, piecewise slopes models, and exponential models.

SAS Syntax for Data Import:

```
* Define global variable for file location to be replaced in code below;
%LET filesave = C:\Dropbox\22_PSQF6271\PSQF6271_Example3b;
* Location for SAS files for these models (uses macro variable filesave);
LIBNAME filesave "&filesave.";

* Import chapter 3 six-occasion stacked data;
DATA work.Example3b; SET filesave.SAS_Chapter3b; RUN;

* Sort by PersonID and session;
PROC SORT DATA=work.Example3b; BY PersonID session; RUN;
```

STATA Syntax for Data Import:

```
// Define global variable for file location to be replaced in code below
   global filesave "C:\Dropbox\22_PSQF6271\PSQF6271_Example3b"

// Import chapter 3 six-occasion stacked data
   use "$filesave\STATA_Chapter3b.dta", clear

// Sort by person and session
   sort personid session
```

R Syntax for Data Import:

```
# Define variables for working directory and data name
filesave = "C:\\Dropbox/22_PSQF6271/PSQF6271_Example3b/"
filename = "SAS_Chapter3b.sas7bdat"
setwd(dir=filesave)

# Import chapter 3 six-occasion stacked data with labels
Example3b = read_sas(data_file=paste0(filesave,filename))
# Convert to data frame as data frame without labels to use for analysis
Example3b = as.data.frame(Example3b)
# Make new variable for session with reference=6 to match other programs
Example3b$session6=relevel(factor(Example3b$session), ref=6)

# Sort data by PersonID and session (needed for correct RCOV matrix)
Example3b = sort_asc(data=Example3b, PersonID,session)
```

SAS, STATA, and R Syntax for Descriptive Statistics:

```
TITLE1 "Chapter 3b Example: Means by session for RT outcome";
* CLASS= stats per session, WAYS=1 per session only;
PROC MEANS NDEC=2 MEAN STDERR STDDEV VAR MIN MAX DATA=work.Example3b;
     CLASS session; WAYS 1; VAR rt;
RUN; TITLE1;

display "Chapter 3b Example: Means by session for RT outcome"
tabulate session, summarize(rt)

print("Chapter 3b Example: Means by session for RT outcome")
describeBy(x=Example3b$rt, digits=2, group=Example3b$session)
```

SAS Output—notice the different means and variances across sessions…

Analysis Variable : rt rt: Response Time in Milliseconds							
session: Occasion (1-6)	N Obs	Mean	Std Error	Std Dev	Variance	Minimum	Maximum
1	101	1961.89	54.68	549.53	301985.34	1055.40	4159.14
2	101	1815.17	50.65	509.07	259150.07	991.29	3954.37
3	101	1750.03	48.07	483.08	233368.00	992.93	4086.14
4	101	1717.80	46.41	466.42	217543.65	930.46	3611.80
5	101	1707.18	45.83	460.54	212097.58	982.77	3290.98
6	101	1672.14	44.13	443.55	196733.17	917.67	3226.00

All models we will examine will have the same model for the means, saturated by session:

$$RT_{ti} = \beta_0 + \beta_1(T1_{ti}) + \beta_2(T2_{ti}) + \beta_3(T3_{ti}) + \beta_4(T4_{ti}) + \beta_5(T5_{ti})$$

The saturated means model predicts the 6 means from 1 intercept (for session 6) + 5 mean differences (1 for each of the 5 other sessions as a binary predictor).

But they will differ in their model for the variance, corresponding to the three kinds of ANOVAs.

SAS, STATA, and R Syntax for Model 1: Saturated Session Means, E-only Variance (BP → no covariance over sessions)

Variance Model: NO covariance or correlation; EQUAL variances across sessions

```
TITLE1 "Eq 3.10: Between-Person Independent ANOVA via SAS MIXED";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = session / SOLUTION CHISQ CL ALPHA=.05 DDFM=Satterthwaite;
     REPEATED session / R RCORR TYPE=VC SUBJECT=PersonID;
     LSMEANS session / DIFF=ALL CL ALPHA=.05;  * All means and mean diffs;
     ODS OUTPUT InfoCrit=work.FitBP;           * Save -2LL for LRT;
RUN; TITLE1;
```

STATA wouldn't let me use Satterthwaite DDF.

```
display "Eq 3.10: Between-Person Independent ANOVA via STATA MIXED"
mixed rt ib(last).session,                                  ///
          || personid: , noconstant variance reml  ///
          residuals(independent,t(session)) dfmethod(residual) dftable(pvalue)
      display "-2LL = " e(ll)*-2  // Print -2LL for model
      estat ic, n(101)            // Information criteria using level-2 N
      contrast i.session, small   // Omnibus F-test for mean diffs
      margins  i.session          // Means per session
      margins  i.session, pwcompare(pveffects) df(600) // Mean differences
      estimates store FitBP       // Save -2LL for LRT
```

```
print("Eq 3.10: Between-Person Independent ANOVA via R GLS (VC R matrix, =LM)")
BP = gls(data=Example3b, method="REML", model=rt~1+factor(session6),
         correlation=NULL) # VC R matrix (no correlation)
print("Show results with -2LL, total leftover variance, and 95% CIs")
summary(BP); -2*logLik(BP); summary(BP)$sigma^2; confint(BP, level=.95)

print("Session means, pairwise mean differences, and omnibus F-test")
emmeans(ref_grid(BP), pairwise~session6, adjust="none"); anova(BP)
```

SAS Output:

Dimensions

Covariance Parameters	1	**still just e in model for variances**
Columns in X	7	**should be 6 but it counts the unidentified 7th one**
Columns in Z	0	**still no U's yet**
Subjects	101	
Max Obs per Subject	6	

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9155.43252939	
1	1	**9155.43252939**	0.00000000

Use this table to get −2LL with enough precision for your HW.

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	236813					
2		236813				
3			236813			
4				236813		
5					236813	
6						236813

The **R matrix** holds the marginal leftover variances and covariances for each session.

So far no covariance is allowed across time, with equal variance across time.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
Session	**ID**	**236813**	**13672**	**17.32**	**<.0001**	**E variance after accounting for means**

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
9155.4	1	9157.4	9157.4	9158.5	9160.0	9161.0

In SAS REML, #parms is the number of variance model parms specifically (1 here).

Solution for Fixed Effects

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept		**1672.14**	**48.4219**	**600**	**34.53**	**<.0001**	**Beta0**
Session	**1**	**289.76**	**68.4789**	**600**	**4.23**	**<.0001**	**Beta1**
Session	**2**	**143.04**	**68.4789**	**600**	**2.09**	**0.0371**	**Beta2**
Session	**3**	**77.8986**	**68.4789**	**600**	**1.14**	**0.2558**	**Beta3**
Session	**4**	**45.6604**	**68.4789**	**600**	**0.67**	**0.5052**	**Beta4**
Session	**5**	**35.0397**	**68.4789**	**600**	**0.51**	**0.6091**	**Beta5**
Session	**6**	**0**	**.**	**.**	**.**	**.**	

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
session	5	600	23.67	4.73	0.0003	0.0003

This multivariate Wald test tells us there is a significant "omnibus" main effect of session (mean diffs).

Least Squares Means

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr > |t|	**Linear combinations:**
Session	1	1961.89	48.4219	600	40.52	<.0001	**Beta0+Beta1**
Session	2	1815.17	48.4219	600	37.49	<.0001	**Beta0+Beta2**
Session	3	1750.03	48.4219	600	36.14	<.0001	**Beta0+Beta3**
Session	4	1717.80	48.4219	600	35.48	<.0001	**Beta0+Beta4**
Session	5	1707.18	48.4219	600	35.26	<.0001	**Beta0+Beta5**
Session	6	1672.14	48.4219	600	34.53	<.0001	**Beta0**

Differences of Least Squares Means

Note these all have the same SE

Effect	session: Occasion (1-6)		session: Occasion (1-6)		Estimate	Standard Error	DF	t Value	Pr > |t|
session		1		2	146.72	68.4789	600	2.14	0.0325
session		1		3	211.86	68.4789	600	3.09	0.0021
session		1		4	244.10	68.4789	600	3.56	0.0004
session		1		5	254.72	68.4789	600	3.72	0.0002
session		1		6	289.76	68.4789	600	4.23	<.0001
session		2		3	65.1377	68.4789	600	0.95	0.3419
session		2		4	97.3759	68.4789	600	1.42	0.1556
session		2		5	108.00	68.4789	600	1.58	0.1153
session		2		6	143.04	68.4789	600	2.09	0.0371
session		3		4	32.2382	68.4789	600	0.47	0.6380
session		3		5	42.8589	68.4789	600	0.63	0.5316
session		3		6	77.8986	68.4789	600	1.14	0.2558
session		4		5	10.6207	68.4789	600	0.16	0.8768
session		4		6	45.6604	68.4789	600	0.67	0.5052
session		5		6	35.0397	68.4789	600	0.51	0.6091

SAS, STATA, and R Syntax for Model 2: Saturated Session Means, $U_{0i} + e_{ti}$ (CS) Variance (WP → constant covariance across sessions + equal variance across sessions)

```
TITLE1 "Eq 3.10: Univariate Repeated Measures ANOVA via SAS MIXED";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = session / SOLUTION CHISQ CL ALPHA=.05 DDFM=Satterthwaite;
     REPEATED session / R RCORR TYPE=CS SUBJECT=PersonID;
     LSMEANS session / DIFF=ALL CL ALPHA=.05;  * All means and mean diffs;
     ODS OUTPUT InfoCrit=work.FitUWP;          * Save -2LL for LRT;
RUN; TITLE1;
* LRT model comparison;
  %FitTest(FitMore=work.FitUWP, FitFewer=work.FitBP);
```

```
display "Eq 3.10: Univariate Repeated Measures ANOVA via STATA MIXED"
mixed rt ib(last).session,                              ///
           || personid: , noconstant variance reml      ///
           residuals(exchangeable,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      display "-2LL = " e(ll)*-2       // Print -2LL for model
      estat ic, n(101)                 // Information criteria using level-2 N
      estat wcorrelation, covariance   // RCOV matrix
      estat wcorrelation               // RCORR matrix
      contrast i.session, small        // Omnibus F-test for mean diffs
      margins  i.session               // Means per session
      margins  i.session, pwcompare(pveffects) df(500) // Mean differences
      estimates store FitUWP           // Save -2LL for LRT
      lrtest FitUWP FitBP              // LRT model comparison (more, fewer)
```

```
print("Eq 3.10: Univariate Repeated Measures ANOVA via R GLS (CS R matrix)")
UWP = gls(data=Example3b, method="REML", model=rt~1+factor(session6),
          correlation=corCompSymm(form=~1|PersonID) )
print("Show results with -2LL, total leftover variance, and 95% CIs")
summary(UWP); -2*logLik(UWP); summary(UWP)$sigma^2; confint(UWP, level=.95)

print("Show R and RCORR matrices for first person in the data")
getVarCov(UWP, individual="101"); corMatrix(UWP$modelStruct$corStruct)[[5]]

print("Session means, pairwise mean differences, and omnibus F-test using correct DDF")
emmeans(ref_grid(UWP), pairwise~session6, adjust="none"); anova(UWP)

print("Show likelihood ratio test comparing model fit: BP vs. Univ WP")
anova(UWP,BP)  # Order of "more", "fewer"
```

SAS Output:

Dimensions	
Covariance Parameters	2 → 2 variance model parameters
Columns in X	7
Columns in Z	0
Subjects	101
Max Obs per Subject	6

Estimated R Matrix for ID 101 → COMBINED U_{0i} + e_{ti} VARIANCE AFTER DIFFERENT MEANS

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**236813**	202677	202677	202677	202677	202677
2	202677	**236813**	202677	202677	202677	202677
3	202677	202677	**236813**	202677	202677	202677
4	202677	202677	202677	**236813**	202677	202677
5	202677	202677	202677	202677	**236813**	202677
6	202677	202677	202677	202677	202677	**236813**

Estimated R Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8559	0.8559	0.8559	0.8559	0.8559
2	0.8559	1.0000	0.8559	0.8559	0.8559	0.8559
3	0.8559	0.8559	1.0000	0.8559	0.8559	0.8559
4	0.8559	0.8559	0.8559	1.0000	0.8559	0.8559
5	0.8559	0.8559	0.8559	0.8559	1.0000	0.8559
6	0.8559	0.8559	0.8559	0.8559	0.8559	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
CS	**ID**	**202677**	**29470**	**6.88**	**<.0001**	**Var(U_{0i}) after accounting for means**
Residual		**34136**	**2158.96**	**15.81**	**<.0001**	**Var(e_{ti}) after accounting for means**

Null Model Likelihood Ratio Test		
DF	Chi-Square	Pr > ChiSq
1	**802.08**	<.0001

Now there is 1 other parameter in the model for the variance besides residual variance

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8353.4	2	8357.4	8357.4	8359.5	8362.6	8364.6

In SAS REML, #parms is the number of variance model parms specifically (2 here).

Solution for Fixed Effects

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept		**1672.14**	**48.4219**	**129**	**34.53**	**<.0001**	**Beta0**
Session	**1**	**289.76**	**25.9993**	**500**	**11.14**	**<.0001**	**Beta1**
Session	**2**	**143.04**	**25.9993**	**500**	**5.50**	**<.0001**	**Beta2**
Session	**3**	**77.8986**	**25.9993**	**500**	**3.00**	**0.0029**	**Beta3**
Session	**4**	**45.6604**	**25.9993**	**500**	**1.76**	**0.0797**	**Beta4**
Session	**5**	**35.0397**	**25.9993**	**500**	**1.35**	**0.1784**	**Beta5**
Session	**6**	**0**	**.**	**.**	**.**	**.**	

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
session	5	500	164.24	32.85	<.0001	<.0001

This multivariate Wald test tells us there is a significant "omnibus" effect of time.

Least Squares Means

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr > |t|	
Session	1	1961.89	48.4219	129	40.52	<.0001	Beta0+Beta1
Session	2	1815.17	48.4219	129	37.49	<.0001	Beta0+Beta2
Session	3	1750.03	48.4219	129	36.14	<.0001	Beta0+Beta3
Session	4	1717.80	48.4219	129	35.48	<.0001	Beta0+Beta4
Session	5	1707.18	48.4219	129	35.26	<.0001	Beta0+Beta5
Session	6	1672.14	48.4219	129	34.53	<.0001	Beta0

Differences of Least Squares Means

Note these all have the same SE (although it's smaller now)

Effect	session: Occasion (1-6)	session: Occasion (1-6)	Estimate	Standard Error	DF	t Value	Pr > |t|
session	1	2	146.72	25.9993	500	5.64	<.0001
session	1	3	211.86	25.9993	500	8.15	<.0001
session	1	4	244.10	25.9993	500	9.39	<.0001
session	1	5	254.72	25.9993	500	9.80	<.0001
session	1	6	289.76	25.9993	500	11.14	<.0001
session	2	3	65.1377	25.9993	500	2.51	0.0125
session	2	4	97.3759	25.9993	500	3.75	0.0002
session	2	5	108.00	25.9993	500	4.15	<.0001
session	2	6	143.04	25.9993	500	5.50	<.0001
session	3	4	32.2382	25.9993	500	1.24	0.2156
session	3	5	42.8589	25.9993	500	1.65	0.0999
session	3	6	77.8986	25.9993	500	3.00	0.0029
session	4	5	10.6207	25.9993	500	0.41	0.6831
session	4	6	45.6604	25.9993	500	1.76	0.0797
session	5	6	35.0397	25.9993	500	1.35	0.1784

BP vs. WP Univariate ANOVA (from *%FitTest* macro):

Does the saturated means, CS variance model fit better than the saturated means, e-only model?

This is the same result as given by the `Null Model Likelihood Ratio Test`

Likelihood Ratio Test for FitBP vs. FitUWP

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitBP	9155.4	1	9157.4	9160.0	.	.	.
FitUWP	8353.4	2	8357.4	8362.6	**802.077**	1	0 → Yes!

SAS, STATA, and R Syntax for Model 3:

Saturated Session Means, Saturated (Multivariate) Variance

(WP → all possible unequal covariances/correlations, unequal variances across sessions)

```
TITLE1 "Eq 3.10: Multivariate Repeated Measures ANOVA via SAS MIXED";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = session / SOLUTION CHISQ CL ALPHA=.05 DDFM=Satterthwaite;
     REPEATED session / R RCORR TYPE=UN SUBJECT=PersonID;
     LSMEANS session / DIFF=ALL CL ALPHA=.05;  * All means and mean diffs;
     ODS OUTPUT InfoCrit=work.FitMWP;          * Save -2LL for LRT;
RUN; TITLE1;
* LRT model comparison;
  %FitTest(FitMore=work.FitMWP, FitFewer=work.FitUWP);
```

SAS uses level-2 *N* for DF with TYPE=UN, but STATA's version is a little bit different.

```
display "Eq 3.10: Multivariate Repeated Measures ANOVA via STATA MIXED"
mixed rt ib(last).session,                          ///
          || personid: , noconstant variance reml   ///
          residuals(unstructured,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      display "-2LL = " e(ll)*-2          // Print -2LL for model
```

```
estat ic, n(101),                    // Information criteria using level-2 N
estat wcorrelation, covariance       // RCOV matrix
estat wcorrelation                   // RCORR matrix
contrast i.session, small            // Omnibus F-test for mean diffs
margins  i.session                   // Means per session
margins  i.session, pwcompare(pveffects) df(100) // Mean differences
estimates store FitMWP               // Save -2LL for LRT
lrtest FitMWP FitUWP                 // LRT model comparison (more, fewer)
```

```
print("Eq 3.10: Multivariate Repeated Measures ANOVA via R GLS (UN R matrix)")
MWP = gls(data=Example3b, method="REML",model=rt~1+factor(session6),
          correlation=corSymm(form=~as.numeric(session)|PersonID), # UN correlations
          weights=varIdent(form=~1|session))                     # Het variances by session
print("Show results with -2LL, total leftover variance using SD multiplier, and 95% CIs")
summary(MWP); -2*logLik(MWP); summary(MWP)$sigma^2; confint(MWP, level=.95)

print("Show R and RCORR matrices for first person in the data")
getVarCov(MWP, individual="101"); corMatrix(MWP$modelStruct$corStruct)[[5]]

print("Session means, pairwise mean differences, omnibus F-test")
emmeans(ref_grid(MWP), pairwise~session6, adjust="none") # tried mode="df.error"
print("Error because of Satterthwaite DDF, so had to switch to incorrect residual DDF")
lsmeans(MWP, "session6", mode="df.error"); joint_tests(MWP, mode="df.error")

print("Show likelihood ratio test comparing model fit: Univ WP vs. Multiv WP")
anova(MWP,UWP)
```

R GLS could not provide correct DDF → error

SAS Output:

Dimensions		
Covariance Parameters	21	→ number of total variance model parameters
Columns in X	7	
Columns in Z	0	
Subjects	101	
Max Obs per Subject	6	

Estimated R Matrix for ID 101 → TOTAL COVARIANCE MATRIX AFTER DIFFERENT MEANS

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	235659	217994	202607	192154	195360
2	235659	**259150**	230217	213232	202092	193268
3	217994	230217	**233368**	205209	196919	188604
4	202607	213232	205209	**217544**	193676	185321
5	192154	202092	196919	193676	**212098**	187840
6	195360	193268	188604	185321	187840	**196733**

Estimated R Correlation Matrix for ID 101 → TOTAL CORRELATION MATRIX AFTER DIFFERENT MEANS

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8424	0.8212	0.7905	0.7593	0.8015
2	0.8424	1.0000	0.9361	0.8981	0.8620	0.8559
3	0.8212	0.9361	1.0000	0.9108	0.8851	0.8802
4	0.7905	0.8981	0.9108	1.0000	0.9016	0.8958
5	0.7593	0.8620	0.8851	0.9016	1.0000	0.9196
6	0.8015	0.8559	0.8802	0.8958	0.9196	1.0000

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
20	925.64	<.0001

Now there are 20 other parameters in the model for the variance besides residual variance

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8229.8	21	8271.8	8273.4	8294.0	8326.7	8347.7

In SAS REML, #parms is the number of variance model parms specifically (21 here).

Solution for Fixed Effects

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept		**1672.14**	**44.1345**	**100**	**37.89**	**<.0001**	**Beta0**
Session	**1**	**289.76**	**32.7000**	**100**	**8.86**	**<.0001**	**Beta1**
Session	**2**	**143.04**	**26.2031**	**100**	**5.46**	**<.0001**	**Beta2**
Session	**3**	**77.8986**	**22.8842**	**100**	**3.40**	**0.0010**	**Beta3**
Session	**4**	**45.6604**	**20.7853**	**100**	**2.20**	**0.0303**	**Beta4**
Session	**5**	**35.0397**	**18.1168**	**100**	**1.93**	**0.0559**	**Beta5**
Session	**6**	**0**	**.**	**.**	**.**	**.**	

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
session	5	100	83.60	**16.72**	<.0001	<.0001

This multivariate Wald test tells us there is a significant "omnibus" effect of time.

Least Squares Means

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr > |t|	
Session	1	1961.89	54.6805	100	35.88	<.0001	**Beta0+Beta1**
Session	2	1815.17	50.6541	100	35.83	<.0001	**Beta0+Beta2**
Session	3	1750.03	48.0684	100	36.41	<.0001	**Beta0+Beta3**
Session	4	1717.80	46.4101	100	37.01	<.0001	**Beta0+Beta4**
Session	5	1707.18	45.8255	100	37.25	<.0001	**Beta0+Beta5**
Session	6	1672.14	44.1345	100	37.89	<.0001	**Beta0**

Differences of Least Squares Means

Note these all have the different SEs now

Effect	session: Occasion (1-6)	session: Occasion (1-6)	Estimate	Standard Error	DF	t Value	Pr > |t|
session	1	2	146.72	29.8209	100	4.92	<.0001
session	1	3	211.86	31.3658	100	6.75	<.0001
session	1	4	244.10	33.6427	100	7.26	<.0001
session	1	5	254.72	35.8456	100	7.11	<.0001
session	1	6	289.76	32.7000	100	8.86	<.0001
session	2	3	65.1377	17.8233	100	3.65	0.0004
session	2	4	97.3759	22.3009	100	4.37	<.0001
session	2	5	108.00	25.7681	100	4.19	<.0001
session	2	6	143.04	26.2031	100	5.46	<.0001
session	3	4	32.2382	20.0232	100	1.61	0.1105
session	3	5	42.8589	22.6091	100	1.90	0.0609
session	3	6	77.8986	22.8842	100	3.40	0.0010
session	4	5	10.6207	20.4625	100	0.52	0.6049
session	4	6	45.6604	20.7853	100	2.20	0.0303
session	5	6	35.0397	18.1168	100	1.93	0.0559

WP Univariate ANOVA vs. WP Multivariate ANOVA (from *%FitTest* macro):
Does the saturated means, UN variance model fit better than the saturated means, CS variance model?

Likelihood Ratio Test for FitUWP vs. FitMWP

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitUWP	8353.4	2	8357.4	8362.6	.	.	.
FitMWP	8229.8	21	8271.8	8326.7	**123.567**	19	0 → Yes!